Determine the origin IP address of queries against a table?
Simplest method is to use a trigger and log the IP address from sys.dm_exec_connections. You'll need a logging table; I've guessed at a couple of columns you'll want, but you may want others.

CREATE TABLE dbo.LoggingTable
(
    LoggingID INT NOT NULL
        CONSTRAINT PK_LoggingTable
        PRIMARY KEY CLUSTERED
        IDENTITY(1,1)
    , AppName SYSNAME
    , HostName SYSNAME
    , ClientNetAddress VARCHAR(255)
    , LogDateTime DATETIME
);
GO

CREATE TRIGGER dbo.triggername
ON dbo.tablename /* name of the target table  */
FOR INSERT, UPDATE, DELETE
AS
BEGIN
  INSERT dbo.LoggingTable(AppName, HostName, ClientNetAddress, LogDateTime) 
  SELECT app_name(), host_name(), client_net_address, SYSDATETIME()
  FROM sys.dm_exec_connections dec
  WHERE dec.session_id = @@SPID;
END
GO


Note that if the update is coming from a connection on the same machine ("The call is coming from inside the house!"), and using shared memory, you'll get &lt;local machine&gt; as the IP address.

Once you've figured it out, don't forget to drop the trigger.

More information about the columns in sys.dm_exec_connections